Visualizing Electromagnetic Fields Using Gnuplot

Somnath Datta
Professor of Physics (Retired), National Council of Educational Research and Training, New Delhi
Res: 656, "Snehalata", 13th Main, 4th Stage, T K Layout, Mysore 570009, India
datta.som@gmail.com; http://sites.google.com/site/physicsforpleasure

(Submitted 30 - 08 -2015)

Abstract

Gnuplot provides an excellent tool for plotting electromagnetic fields from various sources of electric charges and currents, stationary as well as time-varying. Its extraordinary power can be particularly exploited in plotting the propagating electromagnetic field from a localized source, e.g., an *oscillating electric or magnetic dipole*. In this article we have written the explicit commands in Gnuplot which will draw the **E** field lines of the electromagnetic field due to a harmonically oscillating electric dipole aligned with the Z axis. We have plotted the **E** field lines on the XZ plane over a limited region of radius 2.2 wavelengths around the source. We have shown, side by side, the **B** field lines from the same source. However, the **B** lines are coaxial circles around the Z-axis, and hence do not warrant plotting. The plotting of the **E** field involves plotting a 3D "relief map" with *contours* embedded on it, for a certain function $\psi(x, y, t)$, with t held constant. The contour levels are selected by applying certain criteria. The plotted contours are converted to *directed* contours, by drawing arrowheads on them, indicating the direction of the **E** field, so that they qualify as field lines. We have demonstrated two alternative methods of adding this qualification, namely (a) *planting* the **E** field vectors at selected points on the XZ plane, and extrapolating them to the contour lines; (b) plotting the E_θ component along the X axis, its positive value implying **E** pointing in the negative Z direction and vice versa, and then following this direction around the entire contour. We have worked out method (a) only for $t = 0$; and method (b) for one full cycle of oscillation corresponding to eight values of t spaced at equal intervals. Looking at these eight plots sequentially one sees how the electromagnetic field is propagating across space. We have plotted E_θ along the X axis for the same eight values of t, all of them on the same graph, to get a clear view of how the field is oscillating and moving, and its amplitude is falling as the inverse of the distance. At the end we have used Gnuplot to plot the **E** and **B** fields for a linearly polarized plane electromagnetic wave.

1 Introduction

Distributions of stationary charges and steady currents create *time-independent* fields $\mathbf{E}$ (electric field) and $\mathbf{B}$ (magnetic field), respectively. On the other hand, moving charges, or localized distributions of *time-varying* electric charges and electric currents create both $\mathbf{E}$ and $\mathbf{B}$. We shall represent these fields as $\mathbf{E}(\mathbf{r}, t)$ and $\mathbf{B}(\mathbf{r}, t)$, where t stands for time, and $\mathbf{r}$ for (x, y, z) in the Cartesian coordinate system, (r, θ, ϕ) in the spherical coordinate system, and represents the coordinates of an arbitrary *field point* (rather, the radius vector to the field point.) We usually call these combined fields $(\mathbf{E}, \mathbf{B})$ the *electromagnetic field*.

There are standard formulas for the electromagnetic field for simple cases: (i) A *plane wave* $(\mathbf{E}, \mathbf{B})$ propagating through a limited region of space located far away from the source, and (ii) the Electromagnetic field $(\mathbf{E}, \mathbf{B})$ originating from a *localized source*, namely, a *harmonically oscillating electric dipole, or magnetic dipole*, the mathematical expressions for $(\mathbf{E}, \mathbf{B})$ valid for all space, with the origin of the coordinate system located at the source point (i.e., where the point-like dipole is located.) In this article our interest is limited to these two examples.

The mathematical formulas for these two fields are simple, in fact deceptively simple for the case (ii). We have seen them umpteen number of times in text books, but failed to visualize how the field *looks* like, how exactly it is spread over space, how its angular distribution changes as we go from the near region to the far region. One has to see plots of these fields, to comprehend them, and get a picture. There are articles, and some modern books where one can find plots in which the field lines have been displayed beautifully. Even then, the examples seen by us are deficient in one respect, absence of 'arrowheads' indicating whether the field is pointing 'up or down' along the field lines.

Partly to remove this deficiency, partly to equip the reader with his own tool for drawing the fields, and partly to continue our efforts to clarify concepts in Classical Electrodynamics with illustrated examples, graphics and plots[1, 2, 3], that we are writing this article.

All our plotting work is based on freely available software: (i) Gnuplot (Version 4.6), and (ii) Xfig (version 3.2.5c). One can get them and their Manuals[4] from the internet for licence free use. The operating system is Linux. We have used Debian distribution (version 5.0 Lenny) on our desktop computer, and Mint distribution on our laptop computer.

The author had used Gnuplot extensively in his latest book[5] *Mechanics*, in which he had also written an *Appendix* to introduce the reader to Gnuplot and its applications, namely drawing orbits of planets and space vehicles, trajectories of various objects, plotting coordinate vs time of particles in motion. The reader can benefit from these examples, can use them as a starting point for learning and practising Gnuplot. However, these experiences proved to be inadequate for drawing the field lines of the $(\mathbf{E}, \mathbf{B})$ fields discussed in this article. It required two months of self training, working through a large variety of exercises using two resource books, namely the Gnuplot Manual[4] and Gnuplot Cookbook[6], before the author gained confidence in writing the commands to create the field lines and 'plant' the $\mathbf{E}$ vectors along the field lines.

The objective of this article is twofold: (1) To present a graphical illustration of how electromagnetic field propagates in space (i) from a central localized source, viz., an oscillating dipole, and (ii) as a linearly polarized plane wave; (2) How to use Gnuplot for replicating the same graphs on the reader's own computer.

We have placed greater emphasis on objective (2). For this purpose we have copied the actual commands from the Console, presented them in 14 *Exercises*, starting each one with a heading preceded by a serial number in bold italics, e.g., ***Ex.1, Ex.2***, etc. These exercises can be used to replicate *all* the graphs shown in this article. We have provided explanation of some of the commands, up to Ex.8, in two ways: (a) its meaning at the end of the command line, separated by the symbol #, (b) general explanations/instructions at the end of the Exercise, emphasizing them with the "bullet" symbol "• ". For understanding the unexplained commands, the reader should look them up in the references just cited, in particular the Gnuplot Manual.

It is hoped that experience gained by doing the exercises shown in this article will be found useful by students and teachers for application in a variety of other problems and assignments in physics.

We shall begin with the Electromagnetic field $(\mathbf{E}, \mathbf{B})$ originating from an oscillating Electric dipole, which is more interesting. This work will cover most of this article, and is spread over Sections 2-10 (Pages 4 - 33).

2 The $(\mathbf{E}, \mathbf{B})$ Field of an Oscillating Electric Dipole

Fig.1(a) gives a schematic picture of an oscillating electric dipole. This figure also explains the spherical coordinates used, and the unit vectors associated with them. The observation point P (we shall call it *field point*) is located at the radius vector $\mathbf{r}$, has polar coordinates (r, θ, ϕ). It should be remembered that, unlike Cartesian unit vectors $\mathbf{i}, \mathbf{j}, \mathbf{k}$, the spherical unit vectors $\mathbf{e}_r, \mathbf{e}_\theta, \mathbf{e}_\phi$ should be drawn at P (its components in the directions of the XYZ axes are functions of the angular coordinates θ, ϕ of P.) They are pointing in the directions in which the respective coordinates are increasing. In particular the unit vector $\mathbf{e}_\phi$ lies in a plane parallel to the XY plane and is tangent to a circle with centre on the Z axis. We have brought it down from P to its projection N on the XY plane, and shown it separately in Fig (c), in a reduced scale.

The dipole consists of two metallic domes A and B, each spherical in shape, mounted on a metallic pole of length ℓ. The system is neutral as a whole, but has equal and opposite charges on the opposite domes. If at some instant of time t the sphere A has charge $q_a(t)$, then at the same instant of time the other sphere has charge $q_b(t) = -q_a(t)$. A current $I(t)$ flowing through the pole, driven by an oscillating voltage source (Fig.(b)), will make the charge move back and forth between the domes. It is assumed that this oscillation is taking place harmonically at the angular frequency ω. As a consequence the system will develop a harmonically *oscillating electric dipole moment* $\mathbf{p}(t) = p(t)\mathbf{e}_z = q_a(t)\ell\,\mathbf{e}_z$. Here we have used the symbol $\mathbf{e}_z$ to mean a unit vector in the Z-direction. The *wave vector* $\mathbf{k}$ is defined as

$$\mathbf{k} = k\mathbf{e}_r; \quad \text{where} \quad k = \frac{\omega}{c} = \frac{2\pi}{\lambda} \tag{1}$$

to be called the *wave number*, and λ is the wavelength of the ensuing radiation.

We shall write the "source quantities" in a proper form. We shall assume that the current through the metallic pole is uniform[1] (but time varying). Let q_0 be the maximum charge collecting on each dome. Then

$$\begin{aligned} &q_a(t) = q_0 \cos\omega t; \\ &q_b(t) = -q_a(t) = -q_0 \cos\omega t; \\ &\mathbf{p}(t) = q_0\ell \cos\omega t\,\mathbf{e}_z = p_o \cos\omega t\,\mathbf{e}_z; \\ &\text{where } p_o = q_0\,\ell \end{aligned} \tag{2}$$

is the scalar amplitude of this electric dipole moment. The $(\mathbf{E},\mathbf{B})$ field from this oscillating dipole, at some point P far away from the dipole, and located at spherical coordinates (r,θ,ϕ) is given by the following formulas [7, 8, 9], assuming that $r \gg \ell$.

$$\begin{aligned} \mathbf{E}(\mathbf{r},t) &= \frac{p_o}{4\pi\epsilon_0 r^3}\left[\{\cos(kr-\omega t) + kr\sin(kr-\omega t)\}\,2\cos\theta\,\mathbf{e}_r \right. \\ &\quad \left. + \left\{(1-k^2r^2)\cos(kr-\omega t) + kr\sin(kr-\omega t)\right\}\sin\theta\,\mathbf{e}_\theta\right]. \quad (a) \\ c\mathbf{B}(\mathbf{r},t) &= \frac{p_o}{4\pi\epsilon_0 r^3}\left[\left\{kr\sin(kr-\omega t) - k^2r^2\cos(kr-\omega t)\right\}\sin\theta\,\mathbf{e}_\phi\right]. \quad (b) \end{aligned} \tag{3}$$

Here $(\mathbf{e}_r, \mathbf{e}_\theta, \mathbf{e}_\phi)$ are unit vectors associated with the polar coordinates (r,θ,ϕ), i.e., pointing in the directions of the increments of the respective coordinates.

Note that we have multiplied the magnetic field $\mathbf{B}$ with c, the speed of light in vacuum, to get the modified field $c\mathbf{B}$, which has the same unit as $\mathbf{E}$ in the SI units (volt/m) (and which becomes necessary to express the EM field in a relativistically covariant manner.) Moreover $E = cB$ in the radiation zone, i.e., far away from the source, i.e., regions for which

$$kr \gg 1, \quad \text{or,} \quad r \gg \lambda. \tag{4}$$

In this radiation zone approximation, or *far*

[1] This assumption will be valid for the special case of our investigation in which the wavelength λ of the resulting em wave is much much larger than the length of the rod, i.e., $\lambda \gg \ell$.



Figure 1: *(a) Oscillating Electric Dipole at the origin, and the field point P at spherical coordinates* (r, θ, ϕ); (b) The oscillator driving the dipole; (c) Explaining the unit vector $\mathbf{e}_\phi$, by projecting it on the XY plane.

zone approximation (4), the EM field takes the simple form[**?**]

$$\mathbf{E}(\mathbf{r},t) = -\frac{k^2 p_o}{4\pi\epsilon_0}\frac{\cos(kr-\omega t)}{r}\sin\theta\,\mathbf{e}_\theta.$$
$$c\mathbf{B}(\mathbf{r},t) = -\frac{k^2 p_o}{4\pi\epsilon_0}\frac{\cos(kr-\omega t)}{r}\sin\theta\,\mathbf{e}_\phi. \tag{5}$$

This is the radiation field repesenting an electromagnetic wave emanating from a point like dipole source located at the origin and propagating along the direction of the radius vector $\mathbf{r}$ drawn from the origin. We prefer to use the symbol $\mathbf{n}$ to indicate the direction of propagation. Which means that $\mathbf{n} = \frac{\mathbf{k}}{r} = \mathbf{e}_r$. The radiation field is marked by three important characteristics:

$$\begin{array}{lrcl} \text{(I)} & c\mathbf{B} & = & \mathbf{n}\times\mathbf{E}; \\ \text{(II)} & cB & = & E; \\ \text{(III)} & E, cB & \propto & \frac{1}{r}. \end{array} \tag{6}$$

The opposite of the approximation (4) is the *near zone* approximation:

$$kr \ll 1; \quad \text{or,} \quad r \ll \lambda. \tag{7}$$

In this approximation we consider only the zeroth power of kr. Ignoring the first and second power of kr in Eqs.(3) we get

$$\mathbf{E}\left(\mathbf{r},t\right) \approx \left(\frac{p_o}{4\pi\epsilon_0}\frac{2\cos\theta\,\mathbf{e}_r + \sin\theta\,\mathbf{e}_\theta}{r^3}\right)\cos\omega t.$$
$$c\mathbf{B}\left(\mathbf{r},t\right) \approx \mathbf{0}. \tag{8}$$

3 Angular Distribution of the Power Radiated by the Oscillating Dipole

The Poynting's vector, defined as

$$\mathbf{S} = \epsilon_0 c\left[\mathbf{E}(\mathbf{r},t)\times c\mathbf{B}(\mathbf{r},t)\right] \tag{9}$$

gives the flux density of the radiated electromagnetic energy. Consider a point P located far away from the origin, at the spherical coordinates (r,θ,ϕ). The EM field at this point is given by Eq. (5). Therefore, the instantaneous electromagnetic energy flux density at this point is given as

$$\begin{aligned}\mathbf{S}(\mathbf{r},t) &= \epsilon_0 c\left\{\frac{k^2 p_o}{4\pi\epsilon_0}\frac{\cos(kr-\omega t)}{r}\sin\theta\right\}^2(\mathbf{e}_\theta\times\mathbf{e}_\phi)\\ &= \frac{ck^4p_o^2}{16\pi^2\epsilon_0}\frac{\cos^2(kr-\omega t)}{r^2}\sin^2\theta\,\mathbf{e}_r.\end{aligned} \tag{10}$$

The above equation gives the instantaneous value of $\mathbf{S}$. Since the dipole will be oscillating very fast, at frequencies of the order of kHz, what is more relevant is the *time-averaged value* of $\mathbf{S}$, to be written as $<\mathbf{S}>$. This is easily obtained by noting that $<\cos^2(kr-\omega t)>=\frac{1}{2}$. Then,

$$<\mathbf{S}(r,\theta,\phi)>=\frac{ck^4p_o^2}{32\pi^2\epsilon_0}\left(\frac{\sin^2\theta}{r^2}\right)\mathbf{e}_r. \tag{11}$$

The power radiated per unit solid angle is

$$\frac{dP}{d\Omega} = r^2 <\mathbf{S}(r,\theta,\phi)>=\frac{ck^4p_o^2}{32\pi^2\epsilon_0}\sin^2\theta. \tag{12}$$

Plotting the $\sin^2\theta$-angular distribution of the radiated power, as given in Fig. 2, is a trivial application of Gnuplot. The dipole is oriented along the Z axis and is labelled $\tilde{\mathbf{p}}$ (to indicate that it is alternating harmonically). The polar angle θ is measured from the positive Z axis.

We have indicated the strength of the Poynting's vector $\mathbf{S}$ over a sphere of radius r, by the length of the arrow representing this vector. The shaded double-lobe about the Z axis is the XZ plane cross section of an axially symmetrical doughnut like 3-dimensional plot of the radiated power, varying as $\sin^2\theta$, which is characteristic not only of dipole radiations (from both electric and magnetic dipoles), but also of radiation from an accelerating charge (in non-relativistic motion.) In the latter case, the angle θ is measured from the direction of the instantaneous acceleration vector.

It is especially notable that there is *no radiation along the axis of the dipole*, and *maximum radiation along the plane perpendicular to it.*

4 Plotting with Gnuplot

The commands of Gnuplot are to be written on Console. We shall show a few examples of how to use Gnuplot. Let us observe a few conventions. Every command in Gnuplot is preceded by the following prompt



Figure 2: *Energy flow from an oscillating Electric Dipole*

gnuplot>
by which Gnuplot asks us to write a command line. On the other hand when Gnuplot executes a command and gives its answer or response, there is no such prompt. For economy of space we shall write
>
instead of the full prompt "gnuplot>". We shall also use the "comment" symbol '#' for explaining a particular command to the reader. This symbol '#' is analogous to the "comment" symbol '%' used in LaTeX. Gnoplot ignores anything written after # on the same line.

To illustrate these points we shall ask Gnuplot to give the value of a certain function $f(x)$ which is either already there in its own "library", or which we have just defined, for a few specific values of the argument.

We have copied below the "script" from the Console in "footnote size typewriter font", to mark them out from the main text, and to adjust them within the limited space of a column.

Ex.0

```
UrComputer:~ UrDir$ gnuplot # First line
G N U P L O T
    Version 4.6 patchlevel 4 ....
    Build System: Linux i686
    Copyright (C) ....
    Terminal type set to 'wxt'
> f(x)=3*x**2*cos(x)
  # The function f(x) = 3x^2 cos(x) is defined
> print f(1), sin(pi/2), cos(pi/2)
  #  f(x) at x = 1, others at x = π/2.
```

1.62090691760442 1.0 6.12323399573677e-17
values are $1.62090691760442, 1.0,$
$6.12323399573677 \times 10^{-17} = 0$

- The *first line* is the same for all examples to follow. It shows the name of your computer, working directory, and what you have typed just after the $ sign to start Gnuplot.

We shall begin with our first real example.

Ex.1. To plot the angular distribution of the *radiated power* as given in Eq. (12). Here let us note that in the 3-dimensional spherical coordinate system the the polar angle θ is measured from the Z axis, as shown in Fig. 1. In the plane polar coordinate system, on the other hand, the polar angle is measured from the X axis. Since we are going to plot a 2-dimensional curve, the latter coordinate system has to be used and the same function $\sin^2\theta$ written in Eq. (12) has to be written as $\sin^2(\pi/2-\theta) = \cos^2\theta$. We now write the commands.

```
> set polar #  polar coordinates
   dummy variable is t for curves
> set grid polar # draw grid lines
> set trange [0:2*pi] # range of θ
> set rrange [0:1.2] # range of r
> set size square # shape of the plot area
> set xrange [-1.2:1.2]; set yrange [-1.2:1.2]
  # range of x,y coordinates
> plot ((cos(t))**2) # the plot is drawn instantly
> set term fig color portrait size 15, 15
metric pointsmax 1000 solid font "Times-Roman,12"
  # Terminal specification
Terminal type set to 'fig'
Options are 'color small pointsmax 1000 portrait
metric solid textnormal font "Times Roman,12"
linewidth 1 depth 10 version 3.2 size 15 15'
> set title "S4dpol-150623.fig"
  # a label to appear on the plot
> set out "S4dpol-150623.fig"
  # to save as, file name
> replot # replots and saves
```

- The default coordinate system is Cartesian. To plot in the polar system, the command "set polar" is necessary. The function has to be written in the form $r = f(\theta)$. The letter t stands for θ, when plotting "polar".

- Immediately after the command "plot ((cos(t))**2)" the plot appears on the screen, but we are unable to save it in the computer. For this purpose we have to specify a "terminal". The saved plot will look identical with the one now on screen, if we choose "set term png". However, we are more comfortable with the terminal "fig" for editing using Xfig. The command "set term fig ..." not only sets the terminal, but also makes further specifications, e.g, solid and coloured lines for plots as well as grids, font name and size for labelling, size of the screen 15 cm $\times$ 15 cm, etc. The plot is now stored in the working directory with extension ".fig".

- It has been our practice to add today's date (yymmdd) as part of the file name (which the reader need not follow.) We now start Xfig, look for the file name, bring the plot on the screen, for which we shall use the term "canvas". We work on this canvas and do some editing. For example, we give extra labels, draw

X, Z coordinate axes, change the grid lines from "solid" to "dashed", change the colours of the curve, as we feel necessary. See Ex.2 for a better example.

- Note that the "first line" pointed out in Ex.0, and "the terminal specification" in Ex.1 will be common for all Exercises to follow, so that we shall not repeat them any more.

We shall refer to the entire set of commands given in Ex.1 as "script" (term borrowed from Gnuplot Cookbook.) The script written in Ex.1 results in the plot shown in Fig. 3.

5 Plotting the Electric Field from the Oscillating Dipole in the Near Zone

The term "plotting the field" here means plotting the field lines, i.e., imaginary curves in space such that the **E** field at any point in space is tangential to such a curve passing through that point. But why are we restricting ourselves to the **E** field only, ignoring the **B** field? The answer comes from Eqs. (3). The **B** field has only ϕ-component, i.e., it is pointing in the direction of the $\mathbf{e}_\phi$ vector, so that the field lines are coaxial circles around the Z axis (see Figs. 1 (c), 15(b).) In contrast the **E** field has both r and θ components (i.e., having components along $\mathbf{e}_r$ as well as $\mathbf{e}_\theta$ vectors.) They create interesting and beautiful patters in space, which we need to see, appreciate and admire. Also, **B** accompanies **E** everywhere, being comparatively weaker in the near zone, but equallly strong in the radiation zone. Hence the "graph" of the strength of **E** field is also a graph of the strength of **B** field, especially in the radiation zone. See Fig. 13.

We shall first set up the general differential equation in the *spherical polar co-ordinate system* for field lines of **E** and **B** that are symmetrical about the Z-axis (azimuthal symmetry) so that one complete field line, from beginning to end, is confined to an azimuthal plane (ϕ = constant). For convenience of drawing we have taken this plane to be the XZ plane in Fig. 4(a), which shows a part such a field line. The points P and Q are infinitesimally close points on this curve at radius vectors $\mathbf{r}$ and $\mathbf{r} + d\mathbf{r}$ respectively, so that $d\mathbf{r} = dr\,\mathbf{e}_r + r\,d\theta\,\mathbf{e}_\theta$. The lines of **E** or **B** being tangential to the field line at every point, $E_r/E_\theta = dr/r\,d\theta$ for electric field lines; $B_r/B_\theta = dr/r\,d\theta$ for magnetic field lines. Confining ourselves to electric field, let the field lines be represented by the family of curves $C : r = f(\theta, k)$ where k is a constant. It follows that

$$\frac{dr}{r\,d\theta} = \frac{f'(\theta)\,d\theta}{f(\theta)d\theta} = \frac{E_r}{E_\theta}; \quad \text{or,} \quad \boxed{\frac{f'(\theta)}{f(\theta)} = \frac{E_r}{E_\theta}.} \tag{13}$$

Eq. (13) gives the differential equation[8] for azimuthally symmetrical field lines at any point in space where $E_\theta \neq 0$. We shall now use this equation to plot the **E** field from the oscillating dipole for the near zone. We shall plot the field at $t = 0$, so that $\cos\omega t = 1$



Figure 3: *Angular distribution of the energy flow from an oscillating Electric Dipole*

in Eq. (8a). The oscillating dipole is at its peak value and is pointing up i.e., towards the positive Z axis. Also, $E_r/E_\theta = \frac{2\cos\theta}{\sin\theta} = 2\cot\theta$. Hence,

$$\frac{df}{d\theta} = 2\cot\theta\, f. \qquad \text{Or,} \quad \frac{df}{f} = 2\cot\theta\, d\theta. \quad (a)$$

$$\text{By integration} \quad r = f(\theta, k) = k\sin^2\theta. \quad (b) \tag{14}$$

Eq. (14b) gives the family of dipole field lines, different members of this family corresponding to different, positive values of the constant k. In Fig.4(b) we have shown three such curves on each side of the Z-axis. We created them with Gnuplot, added subsequently "arrowheads" manually to indicate the direction of the field at some selected points.

Note that for each value of k there is one



Figure 4: *(a) Plotting field lines; (b) Quasistatic* **E** *lines from an oscillating electric dipole*

pair of symmetrical lobes about the Z axis on the XZ plane (looking somewhat like the wings of a butterfly), representing a dipole field. These contour curves exist symmetrically around the Z axis, covering the vicinity of the oscillating dipole. To evaluate k for a particular field line Γ we have to find the distance D of a point P on the XY plane through which Γ passes. Since $\theta = \pi/2$ for such a point, it follows that $k = D$. We shall now demonstrate the actual plotting of the field lines.

Ex.2. To plot the field lines from the oscillating dipole at $t = 0$. This example also illustrates the use of "do" command and "line type".

```
> set title "Ed-150623.fig"
> set out "Ed-150623.fig"
> Ed(n,t)=0.2*(1+n)*((cos(t))**2)
> do for [n=0:4] {plot Ed(n,t) lt n}
```

- The script written above has been continued from Ex.1.
- The defined function Ed(n,t) has 2 arguments, t for θ and n for the "iteration" number = 0,1,2,3,4 as specified in the next "do" command. It creates five plots, intersecting the X axis at $\pm(0.2, 0.4, 0.6, 0.8, 1.0)$
- The sub-command"lt n" specifies "line type" for each value of n. In term fig, lt 0= black, 1= red, 2= light green, 3= dark blue, 4= magenta. However, we changed the colour of lt 2 from light green to dark green for better visibility,

the grid lines from solid to dashed, their colours from black to blue.

The actual plot is shown in Fig.5.

6 Fieldlines from Oscillating Dipole

Plotting the **E** field as given by Eq. (3) is much more difficult, because in this case the differential equation (13) is not so easy to integrate. However we shall follow the path given by Orfanidis[7] and achieve a wonderful result.

Let us set $P = \frac{k^3 p_o}{4\pi\epsilon_0}$; $\rho = kr$, $\tau = \omega t$. The EM field $(\mathbf{E}, \mathbf{B})$ has only three non-zero components, as seen from Eq. (3):

$$\begin{aligned} E_r &= \frac{P}{\rho^2}\left[\frac{\cos(\rho-\tau)}{\rho} + \sin(\rho-\tau)\right] 2\cos\theta. && (a) \\ E_\theta &= \frac{P}{\rho^3}\left[(1-\rho^2)\cos(\rho-\tau) + \rho\sin(\rho-\tau)\right]\sin\theta. && (b) \\ cB_\phi &= \frac{P}{\rho^3}\left[-\rho^2\cos(\rho-\tau) + \rho\sin(\rho-\tau)\right]\sin\theta. && (c) \end{aligned} \tag{15}$$

We shall plot only the **E** field, ignoring the **B** field, for the reason cited at the beginning of Sec. 5. The **B** field lines on the XY plane are coaxial circles around the Z-axis, as shown later, in Fig.15.

Plotting implies a "picture" of the field with time frozen, i.e., with t held constant. We shall set

$$\Psi(\rho) \equiv \frac{\cos(\rho-\tau)}{\rho} + \sin(\rho-\tau).$$
$$\text{so that } \frac{d\Psi}{d\rho} = -\frac{1}{\rho^2}\left[(1-\rho^2)\cos(\rho-\tau) + \rho\sin(\rho-\tau)\right], \tag{16}$$

with τ held constant. It follows that

$$\begin{aligned} E_r &= \frac{P}{\rho^2}\Psi(\rho)\, 2\cos\theta; \\ E_\theta &= -\frac{P}{\rho}\frac{d\Psi}{d\rho}\sin\theta. \end{aligned} \tag{17}$$

Then from (13)

$$\begin{aligned} &\frac{dr}{d\theta} = r\frac{E_r}{E_\theta}, \Rightarrow \frac{d\rho}{d\theta} = \rho\frac{E_r}{E_\theta} = -\frac{\Psi}{\frac{d\Psi}{d\rho}} 2\cot\theta. \\ &\Rightarrow \frac{d\Psi}{d\theta} = -2\Psi\cot\theta. \end{aligned} \tag{18}$$

This differential equation is similar to the one in Eq. (14a), except for a negative sign on the right side. Solving it we get

$$\begin{aligned} &\Psi(\rho)\sin^2\theta = C \\ &\Rightarrow \left[\frac{\cos(\rho-\tau)}{\rho} + \sin(\rho-\tau)\right]\sin^2\theta = C' \end{aligned} \tag{19}$$

where C' is a constant.

We now return to the original variables and write the above equations of the field lines as

$$\left[\frac{\cos(kr-\omega t)}{kr} + \sin(kr-\omega t)\right]\sin^2\theta = C \tag{20}$$



Figure 5: $\mathbf{E}$ *field lines intersecting the* X*-axis at* $x = \pm(.2, .4, .6, .8, 1.0)$

Before we start plotting let us be careful about what this plotting operation will involve. We are going to plot the $\mathbf{E}$ field lines on a predecided fixed plane Σ which passes through the line of the oscillating dipole, the same as the Z axis. This plane is to be identified as the XZ plane. The field lines will be shown as contour plots of the function that appears on the left side of Eq. (20), for a fixed value of t. The above function will now be considered to be a function of the *Cartesian* coordinates (x, z), with y set to zero. This is because Gnuplot can make contour plots of functions only of Cartesian coordinates.

When Gnuplot plots either a surface, or a series of contours on that susrface, it expects the equation of the surface to be written in Cartesian coordinates as $z = f(x, y)$. For this purpose we shall *change the variable* $z \to y$, denote the left side of Eq. (20) as $\psi(x, y, t)$, and prepare to plot contours on the *surface* defined as:

$$\begin{aligned} z &= \psi(x, y, t) \\ &\stackrel{\text{def}}{=} \left[\frac{\cos(kr - \omega t)}{kr} + \sin(kr - \omega t)\right] \sin^2\theta. \end{aligned} \tag{21}$$

In this change of variables the surface is raised above the XY plane (instead of the XZ plane). The Z axis now represents the function $\psi(x, y, t)$. *Plotting the field lines* for a given value of t is the same as plotting contours on this surface for selected levels, i.e., for selected values of z, with t held constant. The constant C appearing on the right side of Eq. (20) represents one of those "selected evels". After the plotting operation is done, at some convenient point we shall restore the y-coordinate to the status of the z-coordinate.

The contours on this surface are now given by the equation

$$\begin{aligned} z &= \psi(x, y, t) = \\ &\left[\frac{\cos(kr - \omega t)}{kr} + \sin(kr - \omega t)\right] \sin^2\theta = C. \end{aligned} \tag{22}$$

For every time $t > 0$ and every real value of and C lying within $z_{\max}$ and $z_{\min}$ there is a contour .

Now suppose we take the unit of distance as the wavelength λ, and the unit of time as the time period the harmonic oscillation T, i.e.,

$$\rho = kr = \tfrac{2\pi r}{\lambda} \to 2\pi r;\ \tau = \omega t = \tfrac{2\pi t}{T} \to 2\pi t,$$

Hence,

$$\left[\frac{\cos 2\pi(r - t)}{2\pi r} + \sin 2\pi(r - t)\right] \sin^2\theta = C \tag{23}$$

becomes the equation of the field lines in the polar coordinate sytem.

We shall illustrate this contour plot for the field at $t = 0$, so that

$$\begin{aligned} z &= \psi(x, y) \equiv \psi(x, y, t = 0) \\ &= \left[\tfrac{\cos 2\pi r}{2\pi r} + \sin 2\pi r\right] \sin^2\theta = C. \\ &\text{Alternatively,} \\ &\left[\tfrac{\cos\rho}{\rho} + \sin\rho\right] \sin^2\theta = C. \end{aligned} \tag{24}$$

Gnuplot will not only plot the surface $z = \psi(x, y)$, but show selected contours on this surface, and project them on the XY plane, as we shall demonstrate in the next section.

7 Plotting the E Field at $t = 0$

We shall first draw the field lines at $t = 0$. Note that this instant $t = 0$ is not the begnning of time. The dipole has been oscillating forever, from $t = -\infty$ to $t = +\infty$. However the zero time is taken to be one of those instants when the dipole achieves its peak value, pointing upwards (i.e., in the +Z direction).

For a better view of the details of the field lines limit ourselves to a small region around the origin, within a radius of 2 wavelengths.

It should be remembered that we have chosen the length scale in the unit of the wavelength λ (just before Eq. 23). Therefore our viewing region has a radius of $r = 2$ arond the Z-axis. Before plotting the field lines it will be necessary to decide the values of ψ at which the contours will be drawn. For example we may like the contours to pass through some selected points on the X axis, say, $x = 0.8, 0.9, 1.0, 1.1$. However the contour lines need to be more dense. Hence we shall add more lines. The selection process and the subsequent plot is done in three steps.
Step 1: Selection of the contour levels.

To make the selection we have plotted the $\psi(x, y)$ function along the x-axis, i.e., we have plotted the function

$$\psi(x) \equiv \psi(x, 0) \tag{25}$$

in the range [0.1,2.0], avoiding the origin where the function goes to infinity.

Ex.3. To plot the function $\psi(x)$ along the X-axis.

```
> ro(x)=2*pi*x # defines ρ(x) as in Eq.(23).
> psi(x) = cos(ro(x))/ro(x) + sin(ro(x))
    # defines ψ(x) as in Eq.(24)
> set grid # sets grid lines
> set xtics 0.1; set ytics 0.5; set mytics 5
    # sets tic marks on the X and Y axes
> set xrange [0.1:2]; set yrange [-1.1:2]
    # sets the ranges of x,y values
> set title ‘‘psi(x)-150609.fig"
> set out ‘‘psi(x)-150609.fig"
> plot psi (x)  #    2-D plot
```

The plotted function is shown in Fig.6. It comes with the title “psi(x)-150609.fig”, as per the command given in the 3rd line from the bottom of the command chain. On that plot we have indicated

(1) the values $\psi = -0.89, -0.44, 0.16, 0.70$ corresponding to the following intercepts of the $\psi(x, y)$ function on the x-axis: $x = 0.8, 0.9, 1.0, 1.1$. The *exact* values of ψ can be obtained by writing the command:

```
> print psi(0.8),psi(0.9),psi(1.0),psi(1.1)
-0.889579538602437 -0.444719637070117
0.159154943091895 0.704838937474402
```

(2) zeros of ψ at $x \approx 0.44, 0.98, 1.49, 1.99, ...$
(3) maxima of ψ at $x \approx 1.22, ...$
(4) minima of ψ at $x \approx 0.7, 1.73, ...$
We have shown the above points on the plot“psi(x)-150609.fig”, below the x axis..

We selected the preliminary contour levels at $z = -0.89, -0.44, 0.16, 0.70$ from consideration of (1), and extra levels at $z = 1, 1.3, -0.6, 0.5, 0.85$ from consideration of (2),(3),(4). We have indicated the above values on the left side and on the right side of the plot.

Step 2: Plotting $z = \psi(x, y)$, as a 3D surface, and show the contours. We have achieved the objective through the following commands:

Ex.4. To plot the surface $\psi(x, y)$ and contours on it.

```
> r(x,y) = sqrt(x*x+y*y) # defines r =√(x²+y²)
> st(x,y)=x/r(x,y)
    # defines sin θ
> sr(x,y) = sin(2*pi*r(x,y))
    # defines sin ρ = sin 2πr
> cr(x,y) = cos(2*pi*r(x,y))
    # defines cos ρ = cos 2πr
> psi(x,y) =( cr(x,y)/(2*pi*r(x,y))
  + sr(x,y) )*(st(x,y))**2
    # defines ψ(x,y) as in Eq.(24)
> set size square
    # shape of the plot area
> set surface # surface plot
> set contour both
    # contour on the surface as well as
```



Figure 6: *The function $\psi(x, y)$ drawn along the X-axis, for selection of contour levels*

```
    on the base
> set hidden3d # treat surface as opaque
> set cntrparam levels discrete - 0.89,
  -0.44, 0.16, 0.7, 1, 1.3, -0.6, 0.5, 0.85
    # contour parameters, discrete levels
> set xrange [-2:2]; set yrange [-2:2]
    # domains of x and y
> set xlabel "x-axis"; set ylabel "y-axis"
    # labels on the x and y axes
> set zlabel "psi(x,y)"
    # label on the z axis
> set isosamples 50,50; set samples 10,10
> splot psi(x,y)
    # 3-D surface plot. It is interactive
> set term fig color portrait size 15, 15,
  metric pointsmax 1000 solid font
  "Times-Roman,12"
    # Terminal specification
> set title "psi(x,y)-150610.fig"
    # label the plot
> set out "psi(x,y)-150610.fig"
    # save as, the file name
> replot
```

- The "splot" command, before specification of the terminal, results in a 3-D plot which is *interactive*. You can change the view angle, choose the best perspective, then specify the terminal. Finally when you give the "replot" command, the plot will be saved and seen as you last saw it, before setting the terminal.

The resulting plot is shown in Fig.7

Step 3: Plotting the Field lines of **E**. This is achieved through the following commands, starting with "unset surface" to make sure that the surface will not be drawn again.

Ex.5 Plotting field lines on the XZ plane.



Figure 7: *The isometric plot of the surface function $\psi(x,y)$, the z axis representing the height of the surface above the XY plane. The contours at the selected levels are shown on the surface as well as on the XY plane.*

```
> unset sur
> set view map
> set key bmargin
> set title "psi(x,y)cont-150610.fig"
> set out "psi(x,y)cont-150610.fig"
> splot psi(x,y)
```

- The script written above has been continued from Ex.4.

The resulting plot is shown in Fig.8

8 Planting the E vectors

The **E** field lines drawn in the last section may look impressive, but is deficient on one count. There is no indication of the direction of the field along the lines. One can remedy this defect in one of following two ways.
(a) Actual planting of the **E** vector at selected points of the region under scrutiny and, by superimposing the plot of Fig.8 on the same



Figure 8: *The field lines of* $\mathbf{E}$ *on the* XZ *plane at time* $t = 0$*, due to an oscillating electric dipole* $\tilde{\mathbf{p}}$ *oriented along the* Z *axis, and placed at the origin. We have added a small circular blob, with an arrow piercing through it (not part of the plot), to indicate the location and the direction of the dipole at* $t = 0$.

canvas, get a clear indication of the direction of the field along the field lines;
(b) Make a plot of the value of the transverse component of the field, namely $E_\theta(x)$, along the x axis, which gives the exact value of the field along this axis, since there is no E_r component on the XY-plane. Since $\mathbf{e}_\theta = -\mathbf{k}$ on the XY plane, *positive value of $E_\theta(x)$ implies direction of the $-Z$ axis and vice versa*. Follow these directions along entire field lines.

Each option has its advantage and disadvantage. In option (a) the plotter gets a clear indication of the $\mathbf{E}$ planted all over the space, with its both magnitude and direction in display. However, the procedure is laborious, because it involves creating a "data table" of (x, y, E_x, E_y) for selected points. In comparison, option (b) is relatively quicker and easier to implement.

In this section we shall take up the first option, i.e., option (a).

Gnuplot can "plot vectors". We are using the term "planting vectors" for the same operation. Page 56 of Gnuplot Manual tells us how to do this.

> The 2D **vectors** style draws a vector from (x,y) to (x+xdelta, y+ydelta). The 3D **vector** style is similar, but requires 6 columns of basic data. A small arrowhead is drawn at the end of each vector.
> 4 columns: x y xdelta ydelta
> 6 columns: x y z xdelta ydelta zdelta
> The keyword "with vectors" may be followed by an in-line arrow style specification.
>
> .... plot ... with vectors filled heads

Therefore the operation "planting vectors" begins with preparation of a 4 column data table, in which each row will have four entries, namely the (x,y) componets of the field point, followed by the (E_x, E_y) components of the $\mathbf{E}$ field at that point.

At this point let us note that the formulas given in Eqs. (15) give the (r, θ) components of the $\mathbf{E}$ field. They are to be converted to the (x, y) components of the field by the following formulas.

$$\begin{aligned} E_x &= E_r \sin\theta + E_\theta \cos\theta \\ E_y &= E_r \cos\theta - E_\theta \sin\theta \end{aligned} \tag{26}$$

We shall now carry out this operation through the following steps.

Step 1: Calculating the values of (x, y, E_x, E_y) for selected distances d measured from the origin, as a prelude to the creation of the Data Table.

We have selected d in the range of 0.1 to 1.6, at a fixed interval of 0.1, and their locations at six values of the polar angle, measured from the Z axis, equal to $\theta = n\pi/12$; $n = 1, ..., 6$. The region under observation is the first quadrant of the XZ plane, i.e, $0 \leq \theta \leq \pi/2$; $\phi = 0$. In other words $z > 0$; $x > 0$; $y = 0$; $d = f\lambda = f$, since $\lambda = 1$, and $f = 0.1, 0.2, 0.3, \cdots, 1.6$.

We shall illustrate the procedure here only for $d = 1$ and $d = 0.5$.

The reader should interpret all y coordinates written in the commands below as the z coordinate, following the comments made on page 15.

Step 1A

Ex.6. Calculating the coordinates (x, y) of the selected points.

```
> x(n)=d*sin(n*pi/12); y(n)=d* cos(n*pi/12)
    # x(n), y(n) defined
> d=1   #    d=1
> do for [n=1:6] {print x(n)}
    # values of x(n) for d = 1; n = 1, ..., 6
0.258819045102521
0.5
0.707106781186547
0.866025403784439
0.965925826289068
1.0
> do for [n=1:6] {print y(n)}
    # values of y(n) for d = 1; n = 1, ..., 6
0.965925826289068
0.866025403784439
0.707106781186548
0.5
0.258819045102521
6.12323399573677e-17
> d=.5  #    d=.5
> do for [n=1:6] {print x(n)}
0.12940952255126
0.25
0.353553390593274
0.433012701892219
0.482962913144534
0.5
> do for [n=1:6] {print y(n)}
0.482962913144534
0.433012701892219
0.353553390593274
0.25
0.12940952255126
3.06161699786838e-17
```

Step 1B

Ex.7. Calculating E_x, E_y at the selected points at $t = 0$.

```
> r(x,y)=sqrt(x*x + y*y)
> ro(x,y) = 2*pi*r(x,y);
  ro2(x,y)= (ro(x,y))**2
> ro3(x,y) =(ro(x,y))**3
    # ρ, ρ^2, ρ^3 defined
> sr(x,y)=sin(ro(x,y)); cr(x,y)=cos(ro(x,y))
    # sin ρ, cos ρ defined
> st(x,y)=x/r(x,y); ct(x,y)=y/r(x,y)
    # sin θ, cos θ defined
> Er(x,y) = (cr(x,y)/ro3(x,y)
  + sr(x,y)/ro2(x,y)) * 2 * ct(x,y)
> Et(x,y)=(cr(x,y)/ro3(x,y)
    # (E_r, E_θ) as defined in Eqs.(15)
  + sr(x,y)/ro2(x,y)-cr(x,y)/ro(x,y))*st(x,y)
> Ex(x,y)=Er(x,y)*st(x,y)+Et(x,y)*ct(x,y)
> Ey(x,y)=Er(x,y)*ct(x,y)- Et(x,y)*st(x,y)
    # (E_x, E_y) from Eqs.(26)
> d=1.0 \# d=1
> do for [n=1:6] {print Ex(d*sin(n*pi/12),
  d*cos(n*pi/12))}
    # values of E_x(x,y) for d = 1; n = 1, ..., 6
-0.0367651544198614
-0.0636791154033154
-0.0735303088397228
-0.0636791154033154
-0.0367651544198614
-9.00486573608828e-18
> do for [n=1:6] {print Ey(d*sin(n*pi/12),
  d*cos(n*pi/12))}
  # values of E_y(x,y) for d = 1; n = 1, ..., 6
0.0179140770447072
0.0448280380281612
0.0815931924480226
0.118358346867884
0.145272307851338
0.155123501287745
> d=0.5   #         d=.5
> do for [n=1:6] {print Ex(d*sin(n*pi/12),
  d*cos(n*pi/12))}
0.0553888207210481
0.095936251660179
0.110777641442096
0.0959362516601791
0.055388820721048
1.35663484009156e-17
```

Step 2; Constructing the Data Table

This was done in the following way (1) Start a new text document by invoking Libre Office Writer. (2) Save it as a text document with extension .txt. In this case the name of this file is 'E1vecdataC-150423.txt'. (3) Choose any one of the selected points. Copy-paste the values of x,y from Step 1A, and the values of Ex,Ey from Step 1B (after rounding them off to 2 or 3 decimal places.). The 4 numbers are now displayed not in a row, but in a column, as in

a
b
c
d

(4) take cursor just after the top number in this row (i.e., a), go further by one space, and click del. The next lower number (i.e., b) now comes next to the first number with one blank space in between. In this way bring all the numbers in one row. As a result the copied numbers are now rearranged as
a b c d
Continue this procedure to bring the (x,y,Ex,Ey) values of all selected points in as many rows. The composition of this file is now complete.

The (partial) Data Table below shows the data for the 12 selected points corresponding to $d = 1.0$, 0.5.

```
# x y  Ex  Ey
.26 .97 -.037 .018 # d=1
.5 .87 -.064 .045
.71 .71 -.074 .082
.87 .5 -.064 .118
.97 .26 -.037 .145
1.00 0 0 .155
.13 .48 .06 -0.08 # d=0.5
.25 .43 .1 -.12
.35 .35 .11 -.18
.43 .25 .1 -.23
.48 .13 .06 -.27
.5 0 0 -.29
```

The full data table, saved as **'E1vecdataC-150423.txt'** in the working directory, is shown in the Appendix.

Step 3: Planting the **E** *vectors on the first quadrant of the ZX Plane.*

Ex.8.

```
> set  size square
> set  pointsize 0.5
> set xrange [-0.1:1.7];
  set yrange [-0.1:1.7]
  # sets x-range and z-range
> set xtics 0.5; set mxtics 5
  # sets tic marks along the x-axis
> set ytics 0.5; set mytics 5
  # sets tic marks along the z-axis
> set title ‘‘E1vecB-150505.fig"
  # title that appears at the top
  of the plot
> set out ‘‘E1vecA-150505.fig"
  # filename of the plot, stored in UrDir
> set key bmargin
> plot 'E1vecdataC-150423.txt' using
  1:2 with points pt 7,
  'E1vecdataC-150423.txt' using
  1:2:3:4 with vectors size .02, 15
  filled lt 3
```

- The last command plants (a) the field points from col 1,2 of the data file 'E1vecdataC-150423.txt' with 'point type' 7 (circular blob), (b) the vectors from col 1, 2, 3, 4 of the same data file, as straight lines, starting at these field points, and terminating at an arrowhead. The size of the arrowhead is indicated in the subcommand 'with vectors size .02, 15 filled lt 3' (length = .02,



Figure 9: *The* **E** *vectors on the* XZ *plane at time* $t = 0$*, shown on the first quadrant. They are planted at selected points marked with red dots. These points are distributed along radial lines at equal length intervals of 0.1 from* $d = 0.4$ *to* $d = 1.6$*, and at equal angular intervals of* $\pi/12$ *from* $\theta = \pi/12$ *to* $\theta = \pi/2$*. The oscillating dipole is shown at the origin as* $\widetilde{\mathbf{p}}$

sloping angle of the arrows = 15^0, line type 3.)

- If we want to make the vectors look longer, by increasing their size by, say 50%, the *second part* of the last command should be modified to:

  ```
  ... 'E1vecdataC-150423.txt' using
  1:2:(1.5*$3):(1.5*$4) with vectors
  size .02, 15 filled lt 3.
  ```

  See pp.72,169 of Gnuplot Manual, Example on p.28 of Gnuplot Cookbook.

We have shown the planted **E** vectors in Fig. 9.

Step 4: Plotting the Field lines
Ex.9

```
> r(x,y) = sqrt(x*x+y*y)
> st(x,y)=x/r(x,y)
> sr(x,y) = sin(2*pi*r(x,y));
  cr(x,y) = cos(2*pi*r(x,y))
> psi(x,y)=( cr(x,y)/(2*pi*r(x,y))
  +sr(x,y) )*(st(x,y))**2
> set contour base
> unset surface
> set view map
> set cntrparam levels discrete -0.89,-0.44,
  0.16,0.70,1,1.3,-0.6,0.5,0.85
> set isosamples 80,80
> set title "E1lineD-150505.fig"
> set out "E1lineD-150505.fig"
> splot psi(x,y)
```

- The commands are a continuation of those in Ex.8.
- The contur plots, even though they look 2-dimensional, are to be treated as surface plots (3-dimensional). Hence the "splot" command.

We have shown the field lines in Fig. 10.

Step 5: Superimposing the Planted Vectors on the Field lines This operation does not involve Gnuplot. It is mostly a copy-paste operation done in xfig, and shown in Fig. 11. The canvas is divided into four partitions: (a), (b), (c), (d). We have copied Figs. 10 and 9, scaled down to about half their dimensions, and pasted them in partitions (a) and (b). These two figures are superimposed in partition (c), in which we find the field vectors embedded in the neighbourhood of the field lines. We extrapolate their directions into the field lines, in partition (d), by drawing short tangents along the curve with arrowheads, and get directed field lines.

9 Plotting the E Field at equal time intervals of $T/8$ for one full period T.

We shall now take up the option (b) mentioned at the beginning of Sec.8, not just for $t = 0$, but for eight instants of time, taken at equal intervals spread over one full period of oscillation of the dipole, namely, $t = 0, T/8, 2T/8, 3T/8, 4T/8, 5T/8, 6T/8, 7T/8$. We shall first make two series of plots for each one of the above instants, namely, (i) plot the field lines on the XZ plane over a square region: $[-2.5\,\lambda \le x \le 2.5\,\lambda, -2.5\,\lambda \le z \le 2.5\,\lambda]$; (ii) plot E vs x along the X-axis for the same x-range $-2.5\,\lambda \le x \le 2.5\,\lambda$. The



Figure 10: *The* $\mathbf{E}$ *field lines on the* XZ *plane at time* $t = 0$*, shown on the first quadrant, drawn in the x-range [-0.1:1.7] and z-range [-0.1:1.7]. The oscillating dipole placed at the origin is shown as* $\widetilde{\mathbf{p}}$.



Figure 11: *Drawing directed* **E** *field lines (lines of force) on the* XZ *plane at time* $t = 0$*, on the first quadrant.*



Figure 12: **E** *field lines at t=0.*

directions of the field obtained in (ii) will be extrapolated in (i) along the field lines.

Step 1: Plotting the field lines over one full period at 8 equally spaced instants: $t = nT/4$; $\tau = n\pi/4$; $n = 0, 1, ..., 7$

Ex.10

```
> set size square
> set xrange [-2.5:2.5]; set yrange [-2.5:2.5]
> set xtics 0.5; set mxtics 5
> set ytics 0.5; set mytics 5
> set grid xtics ytics back linetype 3
> unset key
> r(x,y) = sqrt(x*x+y*y)
> st(x,y)=x/r(x,y)
> sr(x,y) = sin(2*pi*r(x,y));
  cr(x,y)=cos(2*pi*r(x,y)) # n = 0
> psi(x,y)=(cr(x,y)/(2*pi*r(x,y))
  +sr(x,y))*((st(x,y))**2)
> set contour base
> unset surface
> set view map
> set cntrparam levels discrete -0.89, -0.44,
  0.16,0.70,1,1.3,-0.6,0.5,0.85
> set isosamples 100,100
> set out "Evstime0-150518.fig"
> splot psi(x,y)
> sr(x,y)=sin(2*pi*r(x,y)-pi/4);
  cr(x,y)=cos(2*pi*r(x,y)-pi/4) # n = 1
> psi(x,y) =( cr(x,y)/(2*pi*r(x,y))
  +sr(x,y))*((st(x,y))**2)
> set out "Evstime1-150518.fig"
> splot psi(x,y)
..........................................
> sr(x,y) = sin(2*pi*r(x,y)-7*pi/4);
  cr(x,y)=cos(2*pi*r(x,y)-7*pi/4) # n = 7
> psi(x,y)=(cr(x,y)/(2*pi*r(x,y))
  +sr(x,y))*((st(x,y))**2)
> set out "Evstime7-150518.fig"
> splot psi(x,y)
```

- The field lines corresponding to $t = 0$ are shown in Fig.12
- The command lines corresponding to n=2,...,6 are not shown.
- The field lines corresponding to $t = nT/4$; $n = 1, ..., 7$ are not shown. However, they have been saved in the working direcory under seven file names as specified above, namely, Evstime1-150518.fig, ... ,Evstime7-150518.fig. They are to be used in the Step 5 below.

Step 2: Plotting E_θ vs x along the X-axis. We shall however go further and plot E_θ, B_ϕ on the same plot only for $t = 0$.

Ex.11 To plot $E_\theta(x)$ and $cB_\phi(x)$ at $t = 0$ along the X-axis.

```
> set xtics 0.5; set mxtics 5
> set ytics 0.15; set mytics 3
> set grid xtics ytics back linetype 3
> set xrange [0.2:2.5];set yrange [-0.3:0.3]
> ro(x) = 2*pi*x; ro2(x)=(ro(x))**2;
  ro3(x) =(ro(x))**3
> sr(x) = sin(ro(x));cr(x) = cos(ro(x))
> Et(x)=cr(x)/ro3(x)+sr(x)/ro2(x)
  -cr(x)/ro(x) # n=0
> Bf(x) = sr(x)/ro2(x)-cr(x)/ro(x)
> set xrange [0.2:2.5];
  set yrange [-0.3:0.3]
> set title "EBvsx0-150621.fig"
> set out "EBvsx0-150621.fig"
> plot [0.2:2.5] [-0.3:0.3] Et(x),Bf(x)
```

We have shown this combined plot in Fig. 13. Note from Eq. (15) that $\mathbf{E}$ has only θ component on the XY plane, and $c\mathbf{B}$ has only ϕ component everywhere. Therefore along the X axis, $E = E_\theta, B = B_\phi$. We have shown in Fig. 13 E and cB vs x on the same graph at $t = 0$. It is seen that the two fields are almost equal for $x > 0.6$.



Figure 13: *E and B vs x along the X-axis, at $t = 0$.*

Ex.12 To plot $E_\theta(x)$ along the X-axis, for $t = nT/8$; $n = 1, ...7$.

```
> sr(x)=sin(ro(x)-pi/4);
  cr(x)=cos(ro(x)-pi/4) # n=1
> Et(x)=cr(x)/ro3(x)+sr(x)/ro2(x)-cr(x)/ro(x)
> set title "Evsxn1-150521.fig"
> set out "Evsxn1-150521.fig"
> plot Et(x)
....................................
> sr(x) = sin(ro(x)-7*pi/4);
  cr(x)=cos(ro(x)-7*pi/4) # n=7
> Et(x)=cr(x)/ro3(x)+sr(x)/ro2(x)-cr(x)/ro(x)
> set title "Evsxn7-150521.fig"
> set out "Evsxn7-150521.fig"
> plot Et(x)
```

- The commands are a continuation of those in Ex.11.
- The command lines corresponding to n=2,...,6 are not shown.
- The seven graphs plotted above are not shown. However, they have been saved in the working direcory under seven file names as specified above, namely, Evsxn1-150521.fig, ... , Evsxn7-150521.fig. These plots are to be used in the Step 5 below.

Step 3: Insert arrows alongside **E** *field lines by finding their directions from the E vs x plots, at $t = 0$.*

We have illustrated in Fig. 14 how this operation has been carried out. We copied two figures, namely (a) Fig. 12, and (b) Fig. 13, on a single canvas in xfig. We removed the plot of B_ϕ, so that (b) has only plot of E_θ. Now we placed (b) beneath (a), scaled down (b) such that its boundary lines (corresponding to $x = 0.2, 2.5$) were aligned with the corresponding tic marks in (a). This was done because the graph (a) has xrange [-2.5:2.5] and the graph (b) has xrange [0.2:2.5]. Next, we drew vertical lines from the following points on the graph in (b) to the X-axis of (a): (i) the maxima, (ii) the minima, and the (iii) zeros.

The unit vector $\mathbf{e}_\theta$ coincides with $-\mathbf{k}$ on the XY plane, as mentioned earlier. Therefore, the maxima correspond to peak values of $\mathbf{E}$ in the *negative* Z direction, the minima to the peak values *positive* Z direction, and the zeros to the centres of the loops formed by the contours. We followed the directions obtained from (i) and (ii) through entire loops of the contours.

Step 4: Obtaining directed field lines of $\mathbf{E}$ *and* $\mathbf{B}$*, at* $t = 0$

Now that we have drawn directed field lines of $\mathbf{E}$ at $t = 0$, we need to see it side by side with the $\mathbf{B}$ field at the same instant of time. For this purpose we have copied directed field lines of Fig. 14 as part (a) of Fig. 15, and drawn the field lines of $\mathbf{B}$ as concentric circles in part (b) of the same figure, as suggested by Fig. 1(c), and Eq. (15c).

It is seen from Fig. 13 that the positive and negative peak values of $\mathbf{E}$ and $c\mathbf{B}$ occur at the same set of points on the X axis. Using this as a guide, we have obtained the directions of $\mathbf{B}$ along the field lines, and have indicated them with arrowheads.

Motion of Field lines

The field lines as drawn in Fig. 15 look static. Actually they are moving lines, expanding outward into space with the speed of light. This dynamic character of the field lines is clearly seen especially in the radiation zone, from Eqs. (5). In this zone we can assign a phase, defined as $\varphi = (kr - \omega t)$, to every field line. Any particular phase φ associated with, say the crest (positive maximum) or trough (negative maximum), is given as $\varphi = a =$ constant. Hence, as t changes, the value of r associated with that phase changes accordingly, satisfying $r = a + \frac{\omega}{k}t = a + ct$.

For a better and analytical understanding of the "motion of the field lines" we need to go back to Eq. (22), in an attempt to giving a meaning to the term. Field lines are contours on the surface $z = \psi(x, y, t)$. Imagine two field lines $\Gamma(t)$ and its time evolution $\Gamma(t + dt)$ in a small time-interval dt. They correspond to the same value of the constant C, and are drawn at times t and $t + dt$ respectively. An imaginary point $\mathrm{P}(t)$ on $\Gamma(t)$ at the coordinates (r, θ, ϕ) moves radially to the point $\mathrm{P}(t + dt)$ on $\Gamma(t + dt)$ at the coordinates $(r + dr, \theta, \phi)$. We may then refer to $\dot{r} = \frac{dr}{dt}$ as the radial velocity of the field line at P. We can then obtain $\dot{r}$ by differentiating r with respect to t in the implicit equation Eq. (22), and get

$$\left[\cos(kr - \omega t) - \frac{\sin(kr-\omega t)}{kr} - \frac{\cos(kr-\omega t)}{(kr)^2}\right]\dot{r} = \left[\cos(kr - \omega t) - \frac{\sin(kr-\omega t)}{kr}\right]c. \tag{27}$$

A few wavelenghts away from the source the third term within the square brackets on the left side vanishes, making $\dot{r} \approx c$.

At this point let us pause for a while, and introspect whether these field lines are in conformity with Maxwell's equations in *free*



Figure 14: *Extrapolating the direction of the* **E** *field on the* XZ *plane from the* $E - x$ *graph plotted along the* x*-axis. Time* $t = 0$.



Figure 15: *The* **E** *field on the* XZ *(Fig.(a)), and the* **B** *field on the* XY *plane (Fig (b)), corresponding to the same instant of time:* $t = 0$*. The* **B** *field lines are concentric circles. The solid circles correspond to the maximum values of* **B** *in anticlockwise (*B_ϕ *positive) and clockwise (*B_ϕ *negative) directions, the dashed circles to zero value of the field.*

space, written below.

$$\begin{array}{llll}\nabla\cdot\mathbf{E}=0. & (a) & \nabla\times\mathbf{E}=-\frac{\partial c\mathbf{B}}{\partial(ct)}. & (b)\\ \nabla\cdot c\mathbf{B}=0. & (c) & \nabla\times c\mathbf{B}=\frac{\partial\mathbf{E}}{\partial(ct)}. & (d)\end{array} \quad (28)$$

The form of the equations is a reminder of the relativistic form (i.e., covariant form) of Maxwell's equations. Eqs. (a),(c) suggest that the field lines of both $\mathbf{E}$ and $\mathbf{B}$ should form closed loops, which is obviously the case for both of them. Eq. (c) (representing Fraday's law) requires that the line integral of E around any such closed loop Γ must equal the negative of the area integral of $\frac{\partial cB}{\partial(ct)}$ over the area enclosed by Γ, remembering that the $c\mathbf{B}$ field, lying on the XY plane, is penetrating the XZ plane perpendicularly. This fact cannot be verified by just by looking at the plot.

Step 5: Draw directed field lines of $\mathbf{E}$ *for one full period* T

We have replicated the operation mentioned in Step 3 ($\mathbf{E}$ field at $t=0$) for the remaining seven values of t, covering one full perod, namely, $t = nT/8;\ n = 1,...7$. For each one of these instants, we have taken the field lines from Step 1 (Ex.10), placed below it the corresponding plot of E_θ vs x plot from Step 2 (Ex.12). After completing this work leading to the composite figure for each value of t, we scaled down each one of them to about half of their size (i.e., linear dimension) so that the figures corresponding to $t = 0, T/8, 2T/8, 3T/8$ were arranged in Fig. 16 in four quadrants, and the figures corresponding to $t = 4T/8, 5T/8, 6T/8, 7T/8$ were arranged in Fig. 17 in four quadrants.

Looking at successive pictures, corresponding to successive values of t covering on full period, the reader should get a reasonably good idea of how the field is evolving in time, resulting in a propagating wave in all directions along the XZ plane. The same picture holds for all planes passing through the Z axis.

Step 6. Finally, we have plotted E *vx* x, along the X axis, for one full period at eight equal time intervals on a single graph, as shown in Fig. 18, using the following commands.

Ex.13

```
> set xtics 0.5; set mxtics 5
> set ytics 0.05
> set grid xtics ytics back linetype 3
> ro(x) = 2*pi*x;ro2(x)=(ro(x))**2
> ro3(x) =(ro(x))**3
> sr(x) = sin(ro(x)-n*pi/4);
  cr(x)=cos(ro(x)-n*pi/4)
> Er0(x)=(cr(x)/ro3(x) + sr(x)/ro2(x))
> Et0(x) = Er0(x) - cr(x)/ro(x)
> set title "EvsxD-150520.fig"
> set out "EvsxD-150520.fig"
> do for [n=0:7]
  {plot [0.2:2.5] [-0.3:0.3] Et0(x) lt n}
```

The plots give a clear picture of the EM wave along the X axis. It should be remembered that as we go far away from the orign, the amplitudes of the $\mathbf{E}$ and $c\mathbf{B}$ fields fall off as $1/r$ (compared to $1/r^2$ for the Coulomb field), and they become equal in magnitude, but remain perpendiculr to each other. The pattern remains the same in all directions around the Z axis, except for the fact that their amplitude varies as $\sin\theta$, being zero along the Z axis, and maximum along the XY plane.



Figure 16: *Extrapolating the direction of the* $\mathbf{E}$ *field on the* XZ *plane from the* $E-x$ *graph potted along the* x*-axis. Times* $t = 0, T/8, 2T/8, 3T/8$.



Figure 17: *Extrapolating the direction of the* **E** *field on the* XZ *plane from the* $E - x$ *graph plotted along the* x*-axis. Times* $t = 4T/8, 5T/8, 6T/8, 7T/8$.



Figure 18: E *vs* x *at* $t = 0, T/8, 2T/8, 3T/8, 4T/8, 5T/8, 6T/8, 7T/8,$ *covering one full period* T*, giving an indication of how the EM wave propagates in all directions, its amplitude falling off as* $1/r$ *as we go far away from the source.*

10 Plotting a linearly polarized plane EM Wave

We now come to the part (i) of our work, as mentioned at the end of the Introduction, namely the simplest example of a plane EM wave, originating from a source which is far away from the region where it is detected.

For the simplest example we shall take the direction of propagation to be the Z direction, and the direction of the **E** and **B** fields to be in the X and Y directions respectively. Let the wave be a harmonically varying field. Then[**?**]

$$\begin{aligned} \mathbf{E} &= E_0 \cos k(z-ct)\mathbf{e}_x \quad (a) \\ c\mathbf{B} &= E_0 \cos k(z-ct)\mathbf{e}_y \quad (b) \end{aligned} \tag{29}$$

The direction of the **E** vector is called the direction of *polarization*. This can be any direction perpendicular to the direction of propagation. Since in the present example, the propagation direction is the Z direction, the **E** and **B** fields must be on the XY plane.

The example shown in Eq.(29) assumes a *constant direction of polarization*. When this is the situation, the EM wave is said to be *linearly polarized*. There is another special case in which the magnitude of the **E** field is constant, but rotates uniformly, perpendicular to the direction of propagation. Such a propagating field is said to be *circularly polarized*. The present example illustrates a linearly polarized wave for which the diection of polarization is the X direction. We may even call it an X-polarized plane EM wave. We have depicted this wave in Fig. 19(c).

Note that the direction of the Poynting's vector **S** is the same as the Z direction, and the the waveform shown in the figure is moving in bulk with the speed c in the Z direction. Also we have shown the $c\mathbf{B}$ field, as the companion of the **E** field, so as to convey to the reader the equality $E = cB$ which is intended to be portrayed by equal lengths of the two vectors **E** and $c\mathbf{B}$.

To create the picture of the propagating field we first created two primitive plots, shown in Figs. 19(a) and (b), each having the same cosine function plotted on the XZ plane and the YZ plane, and covering time ranges $[-\pi/2 : \pi/2]$, $[\pi/2 : 3\pi/2]$ respectively. Together they covered one full period of the wave. The final picture of the propagating wave shown in Fig. 19(c) was created by the editing operation : copying-pasting, joining, adding arrows, filling with colors, and then extending further over two more periods. The plots in Figs (a) and (b) were created using Gnuplot through the following command.

Ex.14

```
> set parametric
> set urange [-pi/2:pi/2]
> splot u,0,cos(u),u,-cos(u),0,u,0,0
> set term fig color size 27 18 metric
  pointsmax 1000 solid font "Times-Roman,12"
  depth 50
> set title "EMplaneA-150504.fig"
> set output "EMplaneA-150504.fig"
> set key bmargin
> replot
> set urange [pi/2:3*pi/2]
> set title "EMplaneB-150504.fig"
> set key bmargin
```

```
> splot u, 0, cos( u ), u, -cos( u ), 0, u, 0, 0
> set output "EMplaneB-150504.fig"
```

Appendix

The content of the file E1vecdataC-150423.txt
The data below lists x,y,Ex,Ey for d= 1, .8, .6, .4, .5, .7, .9, 1.1, 1.3, 1.5, 1.4, 1.6, 1.3

```
# x y  Ex  Ey
.26 .97 -.037 .018
.5 .87 -.064 .045
.71 .71 -.074 .082
.87 .5 -.064 .118
.97 .26 -.037 .145
1.00 0 0 .155 #   d=1 on x axis
.21 .77 -.042 -.05
.4 .70 -.072 -.029
.57 .57 -.084 .013
.70 .4 -.072 .055
.77 .21 -.042 .085
.8 0 0 .097 # d=.8 on x axis
.16 .58 .011 -.116
.3 .52 .020 -.124
.42 .42 .023 -.136
.52 .3 .020 -.147
.58 .16 .011 -.155
.6 0 0 -.158 #  d=.6 on x axis
.10 .39 .112 .054
.2 .35 .194 -.028
.28 .28 .224 -.140
.35 .2 .194 -.252
.39 .10 .112 -.334
.4 0 0 -.364 #  d=.4 on x axis
.13 .48 .06 -0.08
.25 .43 .1 -.12
.35 .35 .11 -.18
.43 .25 .1 -.23
.48 .13 .06 -.27
.5 0 0 -.29 # d=.5 on x axis
.18 .68 -.02 -.1
.35 .6 -.04 -.08
.49 .49 -.04 -.06
.61 .35 -.04 -.04
.68 .18 -.02 -.02
.7 0 0 -.02 # d=.7 on x axis
.23 .87 -.05 -.02
.45 .78 -.08 .02
.64 .64 -.09 .06
.78 .45 -.08 .11
.87 .23 -.05 .14
.9 0 0 .16 # d=.9 on x axis
.28 1.06 -.02 .03
.55 .95 -.03 .05
.78 .78 -.04 .07
.95 .55 -.03 .08
1.06 .28 -.02 .1
1.1 0 0 .1 # d=1.1 on x axis
.34 1.26 .02 .02
.65 1.13 .03 .01
.92 .92 .04 -.01
1.13 .65 .03 -.03
1.26 .34 .02 -.05
1.3 0 0 -.05 #  d=1.3 on x axis
.39 1.45 .03 -.01
.75 1.3 .04 -.03
1.06 1.06 .05 -.05
1.3 .75 .04 -.08
1.45 .39 .03 -.1
1.5 0 0 -.1 # d=1.5 on x axis
.36 1.35 .028 .005
.7 1.21 .048 -.015
.99 .99 .056 -.043
1.21 .7 .048 -.071
1.35 .36 .028 -.091
1.4 0 0 -.098 # d=1.4 on x axis
.41 1.55 .015 -.017
.8 1.39 .026 .028
1.13 1.13 .03 -.044
1.39 .8 .026 -.059
1.55 .41 .015 -.07
1.6 0 0 -.074 # d=1.6 on x axis
.31 1.16 .003 .034
.6 1.04 .005 .032
.85 .85 .006 .029
1.04 .6 .005 .026
1.16 .31 .003 .024
```



Figure 19: *A linearly polarized plane electromagnetic field propagating in the Z direction*

```
1.2 0 0 .024 # d=1.2 on x axis
```

Acknowledgement

The author received the basic commands shown in Ex.14 (Plane EM Wave) from his former student Bharat M, and is grateful to Prof. Sophocles J.Orfanidis (Rutgers University) for some helpful suggestions.

References

[1] Somnath Datta, *Minkowski's Space Time*, Physics Education, www.physedu.in, Indian Association of Physics Teachers , Vol 29, No. 2 (Apr-Jun 2013). Article No. 2, 50 pages.

[2] Somnath Datta, *Maxwell's Stress Tensor and Conservation of Momentum in Electromagnetic Field*, Physics Education, www.physedu.in, Indian Association of Physics Teachers , Vol 30, No. 3 (Jul-Sep 2014). Article No. 1, 42 pages.

[3] Somnath Datta, *Solution of Maxwell's Equations Illustrated with Examples and Plots*, under preparation.

[4] Thomas Williams & Colin Kelley, *Gnuplot 4.6*, Manual originally prepared by Dick Crawford, 2012, Version 4.6, http://sourceforge.net/projects/gnuplot (2012)

[5] Somnath Datta, *Mechanics*, Pearson Education, New Delhi (2013). 630 pages

[6] Lee Phillips, *Gnuplot Cookbook*, Packt Publishing Ltd., Birmingham B3 2PB, UK..ISBN 978-1-84951-724-9 www.packtpub.com (2012)

[7] Sophocles J. Orfanidis, *Electromagnetic Waves and Antennas*, ECE Department, Rutgers University, NJ 08854-8058 (2014). Web page: www.ece.rutgers.edu/∼orfanidi/ewa. See pp. 721-722.

[8] Paul Lorrain & Dale Corson, *Electromagnetic Fields and Waves*, 2nd Ed, W.H.Freeman and Co., New York - CBS Publishers & Distributors, Delhi (1986). pp. 63, 595-603.

[9] Wolfgang K.H. Panofsky & Melba Phillips, *Classical Electricity and Magnetism,* 2nd Ed, Addison Wesley, Reading, Massachusetts (1964). pp. 257-259.

[10] David J. Griffith, *Introduction to Electrodynamics*, 3rd Ed, Pearson Education, New Delhi (2006). See Sections 11.1, 9.2.